Ritz Cantilever with Axial Force

The objective in this document is the same as another document posted on this website: to study a cantilever subjected to both lateral and axial force. The first case involves axial compression, as shown here:



Input values

The following parameter values, given in N and mm, are employed in the subsequent plots. λ is the fraction of the axial force relative to the buckling load. The cross-section is circular, made of steel, with diameter d:

$$\text{values1} = \left\{P \to \lambda \frac{\pi^2 EI}{(2L)^2}\right\};$$

$$\text{values2} = \left\{EI \to 200\,000 \frac{\pi d^4}{64}\right\};$$

$$\text{values3} = \{L \to 5000,\ d \to 50,\ F \to 100\};$$

For reference, the static displacement due to the lateral force alone is, in mm:

$$\frac{F L^3}{3 EI} \text{ /. values1 /. values2 /. values3 // N}$$

which yields: 67.9061

Energy Method

The problem is here addressed by an energy formulation and Ritz functions instead of the differential equation. In his book on Structural Mechanics, Hjelmstad nicely explains why `1 - Cos[n π x / (2 L)]` is slightly better than `Sin[n π x / (2 L)]` for approximating the transverse displacement. For that reason, the Ritz functions are:

```
wn = an (1 - Cos[n π x / (2 L)]);
```

The internal strain energy is:

```
U = ∫_0^L 1/2 EI D[D[wn, x], x]^2 dx;
```

The potential energy in the axial force is introduced via Green's strain:

```
HfromP = P ∫_0^L 1/2 D[wn, x]^2 dx;
```

The potential energy in the lateral force is:

```
HfromF = F (wn /. x -> L);
```

The total energy is:

```
Π = U - HfromF - HfromP;
```

The principle of minimum potential energy dictates that the variation of the functional Π must be zero:

```
δΠ = D[Π, an];
```

```
sol = Solve[δΠ == 0, an];
```

Sum over all Ritz functions (notice that *n* is 1, 3, 5, etc.)

```
w = Total[Table[(an (1 - Cos[n π x / (2 L)])) /. sol, {n, 1, 21, 2}]];
```

The tip displacement is, at half the buckling load, for comparison with values presented earlier:

```
w /. x -> L /. values1 /. values2 /. values3 /. λ -> 0.5
```

which yields: {134.88}

Plot the displacement for a few axial force levels, observing the significant increase in lateral displacement as the axial force approaches the buckling load:

```
Plot[{w /. values1 /. values2 /. values3 /. λ -> {0, 0.25, 0.5, 0.8}},
 {x, 0, L /. values3}, PlotStyle → Black]
```

which yields:



Plot the bending moment for the same axial force levels, in kNm:

```
M = EI D[D[w, x], x] 10^-6;
```

```
Plot[M /. values1 /. values2 /. values3 /. λ -> {0, 0.25, 0.5, 0.8},
 {x, 0, L /. values3}, PlotStyle → Black]
```

which yields:



In comparison, the bending moment caused solely by the lateral force is, in kNm:

```
F L 10^-6 /. values3 // N
```

which yields: `0.5`

Convergence of the Solution

Now reconsider the axial compression case, first with a single cosine function as the Ritz approximation:

```
w1 = Total[Table[(an (1 - Cos[n π x / (2 L)])) /. sol, {n, 1, 1, 2}]];
```

Then two terms:

```
w2 = Total[Table[(an (1 - Cos[n π x / (2 L)])) /. sol, {n, 1, 3, 2}]];
```

Then three terms:

```
w3 = Total[Table[(an (1 - Cos[n π x / (2 L)])) /. sol, {n, 1, 5, 2}]];
```

Those solutions are here compared with the previous solution, which employed 10 terms, for the case of compression equal to half the buckling load:

```
Plot[{w1, w2, w3, w} /. values1 /. values2 /. values3 /. λ -> 0.5,
 {x, 0, L /. values3}, PlotStyle → {Black, Thin}]
```

which yields:



The same is done for the corresponding bending moment values, again in kNm:

```
M1 = EI D[D[w1, x], x] 10^-6;
M2 = EI D[D[w2, x], x] 10^-6;
M3 = EI D[D[w3, x], x] 10^-6;
```

```
Plot[{M1, M2, M3, M} /. values1 /. values2 /. values3 /. λ -> 0.5,
 {x, 0, L /. values3}, PlotStyle → {Black, Thin}]
```

which yields:



Tension

Now consider the cantilever shown below, where the lateral force, *F*, is specified as a fraction of the axial force, *P*, having the effect of keeping the angle of the single load shown below constant.



That redefinition is captured in the following revised input:

```
values3 = {L -> 5000, d -> 50, F -> λ 100};
```

Repeat the calculations, but with flipped sign on the potential energy related to the axial force, in order to study tension instead of compression:

```
Π = U - HfromF + HfromP;
```

Principle of minimum potential energy:

```
δΠ = D[Π, an];
```

```
sol = Solve[δΠ == 0, an];
```

Sum over Ritz functions:

```
w = Total[Table[(an (1 - Cos[(n π x)/(2 L)])) /. sol, {n, 1, 21, 2}]];
```

It no longer makes sense to plot solutions for λ=0, because that means zero load. However, we can now increase the load beyond λ=1 because the axial force has a stabilizing effect, in contrast with the compressive buckling situation. The displacements plotted below shows that the lateral displacement increases when the load is initially applied. However, as the load increses beyond reasonable values, the column straightens and the slope of its tip aligns with the slope of the external load.

```
Plot[w /. values1 /. values2 /. values3 /. λ -> {0.25, 0.5, 1, 2, 5, 7, 9},
  {x, 0, L /. values3}, PlotStyle → Black]
```

which yields:



That straightening of the column, with curvature concentrating at its base, is reflected in the corresponding moment diagrams:

```
M = EI D[D[w, x], x] 10^-6;

Plot[M /. values1 /. values2 /. values3 /. λ -> {0.25, 0.5, 1, 2, 5, 7, 9},
  {x, 0, L /. values3}, PlotStyle → Black]
```

which yields: